

SCIENCE & VIE

MENSUEL - N° 874

JUILLET 1990



● "Vaches folles" :
la France aussi ?

● Le Mondiale
en TVHD

MYSTÈRE DANS LE SOLEIL

M 2578 - 874 - 20,00 F



Un générateur de graphiques pour Amstrad

INFORMATIQUE AMUSANTE

Le programme que nous vous proposons ce mois-ci a pour vocation de vous aider à produire vos caractères graphiques sur Amstrad. En effet, une telle opération est souvent fastidieuse à effectuer manuellement; la valeur de chaque groupe de point composant le dessin devant être calculée en binaire puis traduite en décimale en fonction des emplacements qu'ils occupent sur un octet, et ce

pour chaque ligne. De plus, l'Amstrad comportant trois modes d'affichage différents, le résultat peut dans bien des cas être tout autre que celui escompté en raison de la modification des proportions du tracé. Notons enfin que, dans de nombreux cas, le dessin souhaité ne peut tenir sur la grille de base que fournit un unique caractère. Pour notre programme, il sera possible de dessiner sur quatre caractères à la fois. Si, toutefois, ce nombre reste encore insuffisant, il faudra décomposer le dessin de base en plusieurs éléments, qui seront, par la suite, assemblés entre eux. Notons qu'avant chaque calcul de tracé le programme demandera le nombre de cases de base-caractère que vous souhaitez utiliser pour sa réalisation, ainsi que le mode d'affichage qui sera utilisé lors de son emploi; cela afin que les proportions programmées soient respectées.

Ces quelques points précisés, passons à l'écriture de notre programme. Afin de ne pas modifier le fonctionnement normal du clavier, l'ensemble des symboles graphiques seront redéfinis après 128; d'où l'instruction SYMBOL AFTER 128. Le premier symbole graphique redéfini sera d'ailleurs celui servant à préciser

le premier symbole graphique redéfini sera d'ailleurs celui servant à préciser

```

10 CLEAR:MODE 1
20 REM *****
30 REM *
40 REM *   CHOIX DU MODE D'AFFICHAGE   *
50 REM *
60 REM *****
70 LOCATE 10,5: PRINT "BONJOUR,"
80 LOCATE 1,8: PRINT "SOUS QUEL MODE DESIREZ
   VOUS TRAVAILLER"
90 LOCATE 10,10: PRINT "0, 1 OU 2."
100 LET K$=INKEY$
110 IF K$<>"0" AND K$<>"1" AND K$<>"2" THEN
GOTO 100
120 LET M=(ASC(K$))-48
130 MODE M
140 REM *****
150 REM *
160 REM *   CREATION DES TABLEAUX DE
170 REM *   MEMORISATION DES POINTS.   *
180 REM *
190 REM *****
200 DIM A(16,16):DIM R(8,4)
210 REM *****
220 REM *
230 REM *   CREATION DES SYMBOLES
240 REM *   DE LA GRILLE DE TRAVAIL    *
250 REM *
260 REM *****
270 SYMBOL AFTER 127
280 SYMBOL 128,255,129,129,129,129,129,129,2
55
290 SYMBOL 129,255,255,255,255,255,255,255,2
55
300 REM *****
310 REM *
320 REM *   CHOIX DU NOMBRE DE SYMBOLES *
330 REM *   A GENERER EN MEMETEMPS.    *
340 REM *
350 REM *****
360 LOCATE 1,1: PRINT "GENERATION DE"
370 LOCATE 1,3: PRINT "SYMBOLES GRAPHIQUES."
380 LOCATE 1,8: PRINT "NOMBRE DE CASES"
390 LOCATE 1,10: PRINT "HORIZONTALES (1, 2)"
400 LET K$=INKEY$
410 IF K$<>"1" AND K$<>"2" THEN GOTO 400
420 LET XG=(ASC(K$))-48
430 LOCATE 1,15: PRINT "NOMBRE DE CASES"
440 LOCATE 1,17: PRINT "VERTICALES (1 OU 2)"
450 LET K$=INKEY$
460 IF K$<>"1" AND K$<>"2" THEN GOTO 450
470 LET YG=(ASC(K$))-48
480 CLS
490 REM *****
500 REM *
510 REM *   TRACE DE LA GRILLE DE TRAVAIL *
520 REM *
530 REM *****
540 FOR H=0 TO XG-1
550 FOR V=0 TO YG-1
560 FOR N=1 TO 8
570 FOR M=1 TO 8
580 LOCATE (N+(H*8)),(M+(8*V)):PRINT CHR$(12
8)
590 NEXT M
600 NEXT N
610 NEXT V
620 NEXT H
630 LOCATE 1,18: PRINT "JE SUIS PRET;"
640 LOCATE 1,20: PRINT "+FLECHES="
650 LOCATE 1,21: PRINT " DEPLACEMENT"
660 LOCATE 1,22: PRINT "+ESPACE="
670 LOCATE 1,23: PRINT " POINT BLANC"
680 LOCATE 1,24: PRINT "+ENTER="
690 LOCATE 1,25: PRINT " VALIDATION TRACE."
700 REM *****
710 REM *
720 REM *   SAISIE DU TRACE.
730 REM *
740 REM *****
750 LET YC=1
760 LET XC=1
770 LET K$=INKEY$
780 IF K$="" THEN GOTO 860
790 LET K=ASC(K$)
800 LET XC=XC-(K=243)+(K=242):LOCATE 20,20
810 LET YC=YC+(K=240)-(K=241)
820 IF XC<1 THEN LET XC=1
830 IF YC<1 THEN LET YC=1
840 IF XC>8*H THEN LET XC=XC-1

```

senter sur l'écran notre quadrillage de tracé. Le second un carré blanc utilisé pour marquer les points à remplir. Le premier symbole redéfini par le programme portera donc le numéro 130 et ainsi de suite pour les suivants.

Une première page d'accueil demandera de préciser le mode d'affichage sous lequel vous désirez tracer votre dessin. Il faudra choisir entre 0, 1 ou 2. Ensuite, le programme demandera le nombre de cases composant le tracé ; donc de 1 à 4. Ces renseignements fournis, et après contrôle de leur compatibilité avec les possibilités du programme, la grille de traçage sera affichée sur l'écran. La flèche vers la droite permet de faire avancer le curseur sur le quadrillage. De même, les trois autres flèches, gauche, bas, haut, seront utilisées pour son déplacement

ment dans la direction souhaitée. Dès que l'on souhaitera remplir un point de base, il suffira de taper ESPACE. Si un point a été placé par erreur, il est possible de l'effacer. Pour cela, il suffira de faire de nouveau passer le curseur dessus sans frapper ESPACE. Une fois le tracé terminé, ENTER sera frappé pour lancer son codage. Notons que la case sur laquelle se trouve le curseur lorsque ENTER est demandé sera considérée comme vierge lors du cadrage. Si, pour une raison ou une autre, l'ensemble des points doivent être marqués, signalons que le fait de taper ESPACE sur la dernière case de quadrillage (en bas, à droite) provoque automatiquement la validation du tracé et, cette fois, le point sera considéré comme blanc. En fonction du nombre de cases de base utilisées,

de une à quatre instructions SYMBOL N,X,X,X,X,X,X,X,X seront affichées, où N représente le numéro du symbole correspondant et X les valeurs décimales correspondantes à l'emplacement des points. Parallèlement à cela, le dessin, dans sa totalité, sera présenté, sous le mode choisi, en haut à gauche de l'écran.

La frappe de ce programme ne doit pas poser de problème particulier. Notons cependant qu'il a été étudié pour fonctionner uniquement sur Amstrad 464, 664, 6128, etc. Il ne donnerait aucun résultat sur d'autres machines, le procédé de redéfinition de symboles graphiques étant totalement différent. Néanmoins, nous espérons qu'il vous donnera toute satisfaction pour la conception de dessins, voire d'animations.

Henri-Pierre Penel

```

850 IF YC>8*V THEN LET YC=YC-1
860 LOCATE XC,YC:PRINT CHR$(129)
870 IF K=32 THEN GOTO 950
880 FOR T=1 TO 50:NEXT T
890 LOCATE XC,YC:PRINT CHR$(128)
900 LET A(XC,YC)=0
910 IF K=13 THEN GOTO 1060
920 FOR T=1 TO 20:NEXT T
930 IF K=32 THEN GOTO 950
940 GOTO 770
950 LOCATE XC,YC:PRINT CHR$(129)
960 LET A(XC,YC)=1
970 LET XC=XC+1
980 IF XC>8*H THEN LET XC=1:LET YC=YC+1
990 IF YC>8*V THEN GOTO 1060
1000 LET K=0:GOTO 770
1010 REM *****
1020 REM *
1030 REM * CHOIX DU NUMERO DE SYMBOLE. *
1040 REM *
1050 REM *****
1060 LOCATE 1,19:PRINT "(de 130 a 250)"; CHR
$(7)
1070 LOCATE 1,18:INPUT "SYMBOL AFTER";N
1080 IF N<130 OR N>250 THEN GOTO 1060
1090 FOR I=1 TO 19
1100 FOR J=18 TO 25
1110 LOCATE I,J
1120 PRINT " "
1130 NEXT J
1140 NEXT I
1150 REM *****
1160 REM *
1170 REM * CALCUL DE LA TRANSCRIPTION *
1180 REM * NUMERIQUE DE CHAQUE TRACE. *
1190 REM *
1200 REM *****
1210 FOR V=0 TO YG-1
1220 FOR H=0 TO XG-1
1230 FOR L=1 TO 8
1240 FOR C=0 TO 7
1250 LET R(L,(H+(2*V)+1))=R(L,(H+(2*V)+1))+
2^(7-C)*A((C+1)+(8*H),L+(8*V))
1260 LOCATE C+(H*8)+1,L+(V*8):PRINT "0"
1270 IF A((C+1)+(8*H),L+(8*V))=1 THEN LOCATE
C+(H*8)+1,L+(V*8):PRINT CHR$(128)
1280 NEXT C
1290 NEXT L
1300 LET P=1+H+(2*V):LET NS=N+P-1
1310 REM *****
1320 REM *
1330 REM * AFFICHAGE EN TEMPS REEL DES *
1340 REM * NOUVEAUX SYMBOLES GRAPHIQUES. *
1350 REM *
1360 REM *****
1370 SYMBOL NS,R(1,P),R(2,P),R(3,P),R(4,P),R
(5,P),R(6,P),R(7,P),R(8,P)
1380 LOCATE 1,18:PRINT "LE RESULTAT:"
1390 LOCATE 8+H,20+V:PRINT CHR$(NS)
1400 NEXT H
1410 NEXT V
1420 LOCATE 1,23:PRINT "POUR LES DONNEES"
1430 LOCATE 1,25:PRINT "TAPER UNE TOUCHE."
1440 IF INKEY$="" THEN GOTO 1440
1450 REM *****
1460 REM *
1470 REM * AFFICHAGE DES DONNEES *
1480 REM * NUMERIQUES ET SYNTAXE *
1490 REM *
1500 REM *****
1510 MODE 2
1520 FOR I= 1 TO P
1530 LOCATE 1,(2*I)
1540 PRINT "SYMBOL";N+I;" ";
1550 FOR J=1 TO 8
1560 PRINT R(J,I);
1570 IF J<8 THEN PRINT " ";
1580 NEXT J
1590 NEXT I
1600 REM *****
1610 REM *
1620 REM * REBOUCLAGE DU PROGRAMME *
1630 REM * SUR LA FRAPPE D'UNE TOUCHE. *
1640 REM *
1650 REM *****
1660 LOCATE 1,15:PRINT "POUR GENERER D'AUTR
ES SYMBOLES TAPER UNE TOUCHE."
1670 IF INKEY$="" THEN GOTO 1670
1680 GOTO 10

```